

Pràctiques Integrades

1er de Matemàtiques

Pràctica 21

curs 2002–03

21 Endomorfismes, vectors i valors propis, diagonalització...

Un dels problemes típics quan es treballa amb endomorfismes d'un espai vectorial de dimensió finita és el de determinar bases respecte de les quals la matriu de l'endomorfisme és el més simple possible. Per a poder fer aquest estudi hi juga un paper fonamental la determinació dels valors propis i la dels subespais de vectors propis associats. En el paquet **LinearAlgebra** de Maple es poden trobar funcions per a realitzar totes les operacions necessàries.

21.1 Polinomi característic i valors propis

Per a determinar el polinomi característic d'una matriu M podem realitzar els càlculs a partir de la definició: construint la matriu $M - x\text{Id}$ i demanant el determinat d'aquesta nova matriu.

```
> with(LinearAlgebra):  
> M:=< <1,1,3> | <2,7,-2> | <1,3,-6> >;  
> Mx:= simplify(M-x*IdentityMatrix(3));  
> de:=Determinant(Mx);
```

O bé demanant directament que Maple faci el càlcul i ens retorni directament el polinomi amb la comanda `CharacteristicPolynomial`

```
> CharacteristicPolynomial(M,x);
```

(triar una cosa o l'altre només és una qüestió de gust personal ja que els càlculs que s'han de realitzar internament són essencialment els mateixos).

Per a determinar els valors propis (i la seva multiplicitat) caldrà determinar les arrels del polinomi característic. També es pot fer directament amb la comanda `solve` o amb la comanda específica del paquet **LinearAlgebra** que fa aquesta feina (`Eigenvalues`).

```
> M1:=< <1,1,3> | <-1,7,-2> | <1,3,-6> >;  
> solve(Determinant(M1-x*IdentityMatrix(3)));  
> Eigenvalues(M1);
```

Com podreu veure, la diferència entre els dos resultats és ben poca.

21.2 Vectors propis

Un cop es coneixen els valors propis associats a una matriu A , la determinació dels espais de vectors $\vec{v}$ per als que es verifica alguna de les condicions $A\vec{v} = \lambda\vec{v}$ es redueix a la solució de sistemes d'equacions lineals homogenis, $(A - \lambda)\vec{v} = 0$. Recordeu que la comanda `NullSpace` del paquet **LinearAlgebra** resol aquest tipus de sistemes.

Exemple 21.1

La matriu $A = \begin{pmatrix} -2 & 2 & -3 & -1 \\ 2 & -1 & 2 & 0 \\ 3 & -2 & 4 & 1 \\ -2 & 0 & -2 & -1 \end{pmatrix}$ té per valors propis 1 i -1 tal i com es pot comprovar amb

```
> A:=<<-2|2|-3|-1>,<2|-1|2|0>,<3|-2|4|1>,<-2|0|-2|-1>>;
```

```
> Eigenvalues(A);
```

Els espais de vectors propis amb valor propi 1 o -1 es podran determinar amb

```
> B1:=A-IdentityMatrix(4);
```

```
> E1:=NullSpace(B1);
```

```
> B2:=A+IdentityMatrix(4);
```

```
> E2:=NullSpace(B2);
```

Podreu veure que E1 i E2 contenen cada una un parell de vectors que generen l'espai de vectors propis corresponent (multipliqueu A per cada una de les columnes que surten en E1 o E2 per a comprovar que es tenen, efectivament, vectors propis amb valor propi 1 o -1). A la vista dels resultats es pot afirmar que la matriu A és diagonalitzable i es pot confirmar aquest fet considerant

```
> E:=<op(E1)|op(E2)>;
```

```
> E^(-1).A.E;
```

Tot el procés de calcular els vectors propis associats a cada un dels valors propis d'una matriu té una comanda específica en el paquet **LinearAlgebra** que es crida amb **Eigenvectors**. Si considereu

```
> EE:=Eigenvectors(A);
```

obindreu una seqüència amb dos elements, el primer és una columna amb els valors propis (repetits tants cops com la multiplicitat) i el segon és una matriu que té per columnes les components dels vectors propis ordenades de la mateixa manera que els valors propis (si us hi fixeu, veureu que no surten exactament els mateixos vectors que quan apliqueu **NullSpace**). Ara bé, si feu la multiplicació

```
> EE[2]^(-1).A.EE[2];
```

tornareu obtenir la matriu diagonal equivalent a la matriu A .

Amb tot el que heu vist fins ara, l'única dificultat de l'exercici següent hauria de ser la introducció de la matriu M .

Exercici 21.1

Considereu la matriu

$$M = \begin{pmatrix} 1/2 & -1/2 & -1/2 & -1/2 & -1/2 & 0 & 1/2 \\ 7/2 & 9/2 & 3/2 & 7/2 & -3/2 & -1 & -1/2 \\ -1/2 & -5/2 & 1/2 & -1/2 & 3/2 & -2 & 1/2 \\ -5 & -4 & 0 & -6 & 6 & 4 & -1 \\ 0 & 1 & 1 & -2 & 3 & 3 & -1 \\ -7/2 & -9/2 & -1/2 & -11/2 & 11/2 & 3 & -1/2 \\ -3/2 & -3/2 & 5/2 & -11/2 & 17/2 & 4 & -3/2 \end{pmatrix}$$

a) Calculeu el polinomi característic de M .

b) Determineu els seus valors propis i les multiplicitats corresponents.

- c) Determineu els espais de vectors propis associats a cada un dels valors propis.
- d) Comproveu que la matriu M és diagonalitzable.
- e) Calculeu una matriu C tal que $C^{-1} \cdot M \cdot C$ sigui diagonal.

En aquest altre exercici podreu veure una matriu amb el mateix polinomi característic que la matriu M però que no és diagonalitzable.

Exercici 21.2

Considereu

$$B = \begin{pmatrix} 1/2 & -1/2 & -1/2 & -1/2 & -1/2 & 0 & 1/2 \\ 7/2 & 9/2 & 3/2 & 7/2 & -3/2 & -1 & -1/2 \\ -1/2 & -5/2 & 1/2 & -1/2 & 3/2 & -2 & 1/2 \\ -4 & -3 & 1 & -5 & 6 & 3 & -1 \\ 1 & 2 & 2 & -1 & 3 & 2 & -1 \\ -5/2 & -7/2 & 1/2 & -9/2 & 11/2 & 2 & -1/2 \\ 1/2 & 1/2 & 9/2 & -7/2 & 17/2 & 2 & -3/2 \end{pmatrix}$$

calculeu el seu polinomi característic i comproveu que no és diagonalitzable.

Una de les propietats bàsiques del polinomi característic d'una matriu és que *anul·la* a la matriu corresponent. És a dir, si $p(x)$ és el polinomi característic de la matriu A es compleix $p(A) = 0$. En l'exercici següent es posa de manifest aquest fet en un parell d'exemples.

Exercici 21.3

Considereu les matrius A donades per

$$\begin{pmatrix} 1 & 3 & 4 & -1 \\ 0 & 3 & 2 & 1 \\ -1 & -1 & 2 & 5 \\ 2 & 3 & 1 & -1 \end{pmatrix} \quad \begin{pmatrix} -2 & 3 & 1 & -1 \\ 2 & -1 & 0 & 1 \\ -1 & -2 & 2 & 3 \\ 3 & 3 & 1 & -1 \end{pmatrix}$$

calculeu els seus polinomis característics i comproveu que les anul·len.

D'entre tots els polinomis que anul·len un endomorfisme s'en pot determinar el que té el grau més petit (polinomi mínim). La comanda de Maple `MinimalPolynomial` calcula el polinomi mínim d'una matriu.

Exemple 21.2

Es pot calcular el polinomi mínim de $A = \begin{pmatrix} 0 & -1 & 2 \\ 3 & -4 & 6 \\ 2 & -2 & 3 \end{pmatrix}$ amb

```
> A:= < <0,3,2>|<-1,-4,-2>|<2,6,3> >;  
> pm:=MinimalPolynomial(A,x);  
> Pm:=unapply(pm,x);  
> Pm(A);
```

Observeu que la funció Pm que hem definit s'aplica correctament sobre una matriu i sobre A dona com a resultat la matriu nul·la.

Exercici 21.4

Calculeu el polinomi característic de la matriu A anterior i comproveu que té els mateixos factors irreductibles que el polinomi mínim (només que amb multiplicitats menors).